
استخدام DS3231 RTC Module مع الاردوينو

LCD شاشة وعرض التوقيت عل ”RTC “Real Time Clock هذا المشروع كيفية إستخدام وحدة سنتعلم ف

المونات المطلوبة

Arduino Uno

 

https://geeksvalley.com/tutorial/rtc-arduino/
https://geeksvalley.com/tutorial/rtc-arduino/
https://geeksvalley.com/tutorial/rtc-arduino/
https://geeksvalley.com/product/arduino-uno-r3-atmega328/

RTC – Real Time Clock

 

LCD 16×2

Push Buttons

https://geeksvalley.com/product/rtc/
https://geeksvalley.com/product/lcd-16_2/
https://geeksvalley.com/product/tactile-push-button-switch/

Potentiometer 10K Ω

10k Ohm Resistors

https://geeksvalley.com/product/potentiometer/
https://geeksvalley.com/product/potentiometer/
https://geeksvalley.com/product/10k-resistors/

Breadboard

Wires

LCD الشاشة

تعمل الشاشة ف أحد الحالات التالية:

1- ان تستقبل امر من الارودوينو و تقوم بتنفيذة مثلا: امر مسح الشاشة و امر التهيئة 

lcd.begin(16,2);
lcd.clear();

https://geeksvalley.com/product/full-size-breadboard-830/
https://geeksvalley.com/product/breadboard-jumper-wire/

2- ان تستقبل معلومات من الاردوينو و تقوم بعرضها مثلا : كتابة جملة معينة  

lcd.print("Hello");

DS3231 RTC Module

هو عبارة عن جهاز بسيط يتم توصيلة مع الاردوينو بهدف حساب التوقيت. بمعن أنه يمننا إستخدامه كساعه لمعرفة الوقت.

يستخدم بروتوكول I2C للتواصل مع الاردوينو. وما يميزه أنه يعمل ببطارية خارجية خاصة به مما يتيح له امانية الحفاظ عل حساب
التوقيت وحت عند إنقطاع الهرباء عن لوحة الاردوينو.

يقوم بحساب التوقيت مثل: الثوان والدقائق والساعات والايام والشهور والسنوات بشل تلقائ. حيث انه لا يحتاج لإعادة ضبط بسبب
إختلاف عدد الايام ف بعض الشهور عن الأخرى.

توصيل الدارة

قم بتوصيل الدارة كما هو موضح بالصورة التالية :

: DS3231 RTC توصيل وحدة

Arduino RTC Module

Vcc 5v
Gnd GND

A5 SCL
A4 SDA

تقوم وحدة الRTC بحساب التوقيت و ارساله ال الاردوينو ليتم عرضه عل شاشة LCD ، المفاتيح تستخدم لضبط التوقيت ف البداية
فقط.

البرمجة

قم بتنزيل الود البرمج كاملا من خلال الرابط هنـا .

شرح البرمجة :

.Wire.h تبةو الذي تتيحه لنا الم I2C إستخدام بروتوكول بإستخدام الاردوينو، نحتاج ال RTC Module ن من التواصل معنتم حت

لذلك ف البداية نقوم بإدراج متبة Wire.h ، الت تحتوى عل الدوال اللازمة للتواصل بين الاردوينو و الـ RTC Module .كما نقوم
أيضا بإدراج متبة LiquidCrystal.h الت تحتوي عل الدوال الخاصة بشاشة الـ LCD، و الت سيتم إستخدمها لعرض التوقيت.

#include <Wire.h>
#include <LiquidCrystal.h>

.Push Buttons المشروع والخاصة بالـ تم استخدامها ف نقوم بعد ذلك بتسمية منافذ الأردوينو الت

#define ENTER A2

https://geeksvalley.com/wp-content/uploads/2017/10/rtc.ino_.zip

#define UP A1
#define DOWN A0

بعد ذلك قمنا بتحديد عنوان خاص بالRTC Module و هو 0x68 و قمنا بتعيين متغير له :

#define DS3231_I2C_ADDRESS 0x68

أي وحدة تستخدم برتوكول I2C تمتلك عنوان معين ليتمن الاردوينو من التواصل معه. لان هذا البرتوكول يستطيع التواصل مع عدد
كبير من الاجهزة عل نفس الـProtocol Bus . لذلك كل واحد منها يمتلك عنوان معين ليستطيع الاردوينو التواصل معه بدون التأثير

عل باق الاجهزة الموصلة عل الـ Busـ

يتم الحصول عل العنوان الخاص بل موديول من الـ Datasheet  الخاصه به :

 

 ثم نقوم بإنشاء المتغير الخاص بشاشة الـ LCD، وتحديد الـ Pins الموصوله بينها وبين الأردوينو :

LiquidCrystal lcd(2, 3, 4, 5, 6, 7);

 ونقوم بالإعلان عن المتغيرات المستخدمه ف تخزين حالة الـ Push Buttons، والت ستيم استخدامها ف البداية لضبط التوقيت فقط.
. SetH و SetM كما نقوم أيضا بالإعلان عن متغيرات لحفظ التوقيت الذي سنقوم بضبطه مثل متغير

//variables for button states
int enterState = 0, enterStateLast = 0, upState = 0, upStateLast = 0, downState = 0,
downStateLast = 0;

//variables for ui
boolean blinkOn = true; //visibility of ':' between hour and minutes
boolean setVisible = false; //visibility of the set time ui

//variables for new time
int setM = 0; //users new minute value
int setH = 0; //users new hour value

هذا الموديول يتعامل مع الارقام عل النظامBinary Coded Decimal . أي اننا سوف نحتاج لضبط التوقيت ف البداية إل تحويل
.decToBcd() لذلك نستخدم الدالة . BCD الصورة إل Decimal الأعداد من الصورة

//convert normal decimal numbers to binary coded decimals
byte decToBcd(byte val)
{
  return ( (val / 10 * 16) + (val % 10) );
}

ال BCD لتحويل التوقيت المرسل من الموديول من الصورة ، bcdToDec() وعند قراءة قيمة الوقت من الموديول نستخدم الدالة 
.LCD الشاشة ن من عرضها عللنتم ، Decimal الصورة

//convert binary coded decimal to normal decimal numbers
byte bcdToDec(byte val)
{
  return ( (val / 16 * 10) + (val % 16) );
}

 

 الدالة ()setup نقوم بضبط الإعدادات اللازمة للمشروع، مثل تشغيل بروتوكول I2C و شاشةالـ  LCD و قمنا بضبط المفاتيح
المستخدمة ف ضبط التوقيت كمدخل.

//code that runs once at setup
void setup() {
  //start wire and lcd
  Wire.begin();
  lcd.begin(16,2); //(col, rows)

  //intialize buttons as inputs
  pinMode(ENTER, INPUT);
  pinMode(UP, INPUT);
  pinMode(DOWN, INPUT);
}

 الدالة ()loop نقوم بقراءة الـPush Buttons لضبط التوقيت ف البداية ثم نقوم بالتواصل مع الـRTC Module و نعرض التوقيت
. LCD الشاشة عل

void loop() {
  checkButtons();
  printTime();
}

الدالة ()setRTCTime تقوم بإنشاء اتصال بين الأردوينو و الـ RTC Module بإستخدام متبة Wire.  وظيفة هذه الداله هو ضبط
التوقيت الذي سنقوم بإدخاله بإستخدام الـ Push Buttons. ف البداية تقوم بإرسال عنوان الجهاز الذي تريد الإتصال معه وف حالتنا

هنا هو RTC Module و الذي قمنا بتحديده ف بداية الود.

//set the time and date to the RTC
void setRTCTime(byte second, byte minute, byte hour, byte dayOfWeek, byte
                dayOfMonth, byte month, byte year)
{
  // sets time and date data to DS3231

  Wire.beginTransmission(DS3231_I2C_ADDRESS);
  Wire.write(0); // set next input to start at the seconds register
  Wire.write(decToBcd(second)); // set seconds
  Wire.write(decToBcd(minute)); // set minutes
  Wire.write(decToBcd(hour)); // set hours
  Wire.write(decToBcd(dayOfWeek)); // set day of week (1=Sunday, 7=Saturday)
  Wire.write(decToBcd(dayOfMonth)); // set date (1 to 31)
  Wire.write(decToBcd(month)); // set month
  Wire.write(decToBcd(year)); // set year (0 to 99)
  Wire.endTransmission();
}

. BCD الصورة الموديول نقوم بتحويلها ال بعد ذلك تقوم بإرسال التوقيت الذي تم ضبطه مع ملاحظة أننا عندما نرسل البيانات ال

 Wire.write(decToBcd(minute));

بعد إرسال كل البيانات الخاصة بالتوقيت وه الثوان والدقائق والساعات والأيام والشهور والسنة، نقوم بإنهاء الإتصال ل يقوم
الموديول بمعالجة هذه البيانات ف ضبط التوقيت الخاص به.

الحصول عل تقوم بعمل إتصال مع الموديول و قراءة معلومات التوقيت منه.  وسيتم استخدامها ف readRTCTime() الدالة
التوقيت وعرضه.

وه تعمل بنفس اسلوب الداله السابقة مع اختلاف اننا نقوم بقراءة البيانات من الموديول بدلا من إرسالها له، مع ملاحظة اننا ف هذه
. Decimal الصورة ال BCD الحاله نقوم بتحويلها من الصورة

//read the time and date from the RTC
void readRTCTime(byte *second, byte *minute, byte *hour, byte *dayOfWeek,
                 byte *dayOfMonth, byte *month, byte *year)
{
  Wire.beginTransmission(DS3231_I2C_ADDRESS);
  Wire.write(0); // set DS3231 register pointer to 00h
  Wire.endTransmission();
  Wire.requestFrom(DS3231_I2C_ADDRESS, 7);
  // request seven bytes of data from DS3231 starting from register 00h
  *second = bcdToDec(Wire.read() & 0x7f);
  *minute = bcdToDec(Wire.read());
  *hour = bcdToDec(Wire.read() & 0x3f);
  *dayOfWeek = bcdToDec(Wire.read());
  *dayOfMonth = bcdToDec(Wire.read());
  *month = bcdToDec(Wire.read());
  *year = bcdToDec(Wire.read());
}

LCD الشاشة ثم تقوم بعرضها عل readRTCTime() تقوم بقراءة قيمة التوقيت من الموديول بإستخدام الدالة printTime() الدالة
. HH : MM  الصورة مع اضافة كود مهمته عرض التوقيت عل

//reads the RTC time and prints it to the top of the LCD
void printTime()
{
  byte second, minute, hour, dayOfWeek, dayOfMonth, month, year;
  //retrieve time

  readRTCTime(&second, &minute, &hour, &dayOfWeek, &dayOfMonth, &month, &year);
  //print to lcd top
  lcd.setCursor(0,0);
  lcd.print(" Current: ");
  if (hour<10)
  {
    lcd.print("0");
  }
  lcd.print(hour, DEC);
  if (blinkOn == true)
  {
    lcd.print(" ");
    blinkOn = false;
  }
  else if (blinkOn == false)
  {
    lcd.print(":");
    blinkOn = true;
  }
  if (minute<10)
  {
    lcd.print("0");
  }
  lcd.print(minute, DEC);
  delay(100);
}

الجزء السابق مختص بقراءة قيمة التوقيت الذي يقوم الRTC Module بحسابه بدون تدخل من المستخدم . الجزء التال من الود
.RTC Moduleال مختص بقراءة حالة المفاتيح لضبط التوقيت و ارسالة ال

: تعمل كالتال تقوم بقراءة حالة المفاتيح و الت checkButtons() الدالة

المفتاح المسمUP  يقوم بزيادة عداد الدقائق والساعات.
المفتاح المسمDOWN  يقوم بإنقاص عداد الدقائق والساعات.

.RTC Moduleالـ يقوم بتطبيق التوقيت الذي تم ضبطه وإرساله ال ENTER المفتاح المسم

//checks if buttons are pressed and responds accordingly
void checkButtons()
{
  //check enter
  enterState = digitalRead(ENTER);
  if (enterState != enterStateLast)
  {
    if (enterState == HIGH)
    {
      if (setVisible == true)
      {
        byte second, minute, hour, dayOfWeek, dayOfMonth, month, year;
        readRTCTime(&second, &minute, &hour, &dayOfWeek, &dayOfMonth, &month,
&year);
        setRTCTime(0, setM, setH, dayOfWeek, dayOfMonth, month, year);
      }

      else if (setVisible == false)
      {
        showSet();
      }
    }
  }
  enterStateLast = enterState;
  //check up
  upState = digitalRead(UP);
  if (upState != upStateLast)
  {
    if (upState == HIGH)
    {
      if (setVisible == true)
      {
        addMin();
        printSetTime();
      }
      else if (setVisible == false)
      {
        showSet();
      }
    }
  }
  upStateLast = upState;

  //check down
  downState = digitalRead(DOWN);
  if (downState != downStateLast)
  {
    if (downState == HIGH)
    {
      if (setVisible == true)
      {
        subMin();
        printSetTime();
      }
      else if (setVisible == false)
      {
        showSet();
      }
    }
  }
  downStateLast = downState;
}

الدالة ()showSet تقوم بعرض التوقيت الذي نقوم بضبطه ف أسفل الشاشة قبل إرساله ال الموديول ل لا نقوم بإرسال توقيت
خاط بشل غير مقصود.

//displays the new time interface in the bottom of the LCD
void showSet ()
{
  //update new time variables to current RTC values

  byte second, minute, hour, dayOfWeek, dayOfMonth, month, year;
  readRTCTime(&second, &minute, &hour, &dayOfWeek, &dayOfMonth, &month, &year);
    setH = hour, DEC;
    setM = minute, DEC;

  //pints to the LCD
  lcd.setCursor(0,1);
  lcd.print("   New: ");
  printSetTime();
  setVisible = true;
}

. LCD اسفل الشاشة لعرض التوقيت الذي تم ضبطه ف showSet() تستخدمها الدالة printSetTime() الدالة

//prints the new time values on the bottom of the LCD
void printSetTime()
{
 lcd.setCursor(8,1);
 if (setH<10)
  {
    lcd.print("0");
  }
  lcd.print(setH);
  lcd.print(":");
  if (setM<10)
  {
    lcd.print("0");
  }
  lcd.print(setM);
}

تخزين قيم الدقائق والساعه الت تستخدم ف تقوم بضبط المتغيرات الت ود وهالجزء السابق من ال الدوال التالية مستخدمة ف
. Push Buttonsيقوم المستخدم بضبطها بواسطة ال

و تقوم أيضا بعمل اختبار لعداد الدقائق فإذا وصل العدد ال UP المفتاح تقوم بزياد عداد الدقائق عند الضغط عل addMin() الدالة
59 تقوم بزيادة عداد الساعات .

//adds a minute to new time
void addMin()
{
  if (setM < 59)
  {
    setM++;
  }
  else if (setM == 59)
  {
    setM = 0;
    addHr();
  }
}

الدالة ()subMin تعمل عس عمل الدالة السابقة وه تقوم بإنقاص عداد الدقائق عند الضغط عل المفتاح DOWN وتقوم أيضا
باختبار عداد الدقائق فإذا وصل ال 0 تقوم بإنقاص عداد الساعات.

//subtracts a minute from new time
void subMin()
{
  if (setM > 0)
  {
    setM--;
  }
  else if (setM == 0)
  {
    setM = 59;
    subHr();
  }
}

الدالة ()addHr نقوم بإستدعائها الدالة ()addMin عندما يصل عداد الدقائق ال 59 لتقوم بزيادة عداد الساعات بمقدار واحد.

//adds an hour to new time
void addHr ()
{
  if (setH < 23)
  {
    setH++;
  }
  else if (setH == 23)
  {
    setH = 0;
  }
}

الدالة ()subHr تقوم باستدعائها الدالة ()subMin عندما يصل عداد الدقائق ال 0 لتقوم بإنقاص عداد الساعات بمقدار واحد.

//subtracts an hour from new time
void subHr ()
{
  if (setH > 0)
  {
    setH--;
  }
  else if (setH == 0)
  {
    setH = 23;
  }
}


